/*
* Copyright (c) 2009, Ander Ruiz & Alexander Benito
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *
 * - Redistributions of source code must retain the above copyright notice,
 *   this list of conditions and the following disclaimer.
 * - Redistributions in binary form must reproduce the above copyright notice,
 *   this list of conditions and the following disclaimer in the documentation
 *   and/or other materials provided with the distribution.
 * - Neither the name of Mauricio Aguilar O. nor the names of its contributors
 *   may be used to endorse or promote products derived from this software
 *   without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 *
 * @author: Ander Ruiz y Alexander Benito
 * email  : anderruiz@yahoo.es
 * Date   : May 31 , 2009
 *
 */
package org.transFX.ext.swing;

import java.awt.event.FocusEvent;
import java.awt.event.FocusListener;
import java.awt.event.KeyEvent;
import java.awt.event.KeyListener;
import javafx.ext.swing.SwingComponent;
import javax.swing.JComponent;
import javax.swing.JTextField;
import javax.swing.border.EmptyBorder;
import java.awt.Color;

public class TextField extends SwingComponent
{
    var myComponent: JTextField;

    public var verify: function(text:String);
    public var enterPressed: function();
    public var length: Integer;
    public var editable:Boolean on replace
    {
        myComponent.setEditable(editable);
    }
    public var color:Color on replace
    {
        myComponent.setForeground(color);
    }
    public var backcolor:Color on replace
    {
       myComponent.setBackground(backcolor);
    }
    public var text: String on replace
    {
        myComponent.setText(text);
    }


    public override function createJComponent():JComponent
    {
        myComponent = new JTextField();
        myComponent.setBorder(new EmptyBorder(5,5,5,5));
        myComponent.setOpaque(false);
        myComponent.addKeyListener( KeyListener
        {
            public override function keyPressed(keyEvent:KeyEvent)
            {
                if (keyEvent.VK_PASTE == keyEvent.getKeyCode())
                {
                    myComponent.paste();
                }
                if (keyEvent.VK_ENTER == keyEvent.getKeyCode())
                {
                    enterPressed();
                }
            }

            public override function keyReleased( keyEvent:KeyEvent)
            {
                var pos = myComponent.getCaretPosition();
                text = myComponent.getText();
                myComponent.setCaretPosition(pos);
                if (keyEvent.VK_ENTER != keyEvent.getKeyCode())
                {
                    verify(text);
                }
            }

            public override function keyTyped(keyEvent:KeyEvent)
            {
                length = myComponent.getDocument().getLength();
            }
        }
        );
        myComponent.addFocusListener(FocusListener
        {
            public override function focusGained(focusEvent:FocusEvent)
            {
                
            }

            public override function focusLost(focusEvent:FocusEvent)
            {
                verify(text);
            }

        });

        return myComponent;
    }

}
